

# Diffing Mutually Recursive Types

A code tour

Victor Cacciari Miraldo

University of Utrecht

November 25, 2016

## 1 Our Universe

The universe we are using is a variant of Regular types, but instead of having only one type variable, we handle  $n$  type variables. The codes are description of regular functors on  $n$  variables:

```
data Un (n : ℕ) : Set where
  I      : Fin n      → Un n
  K      : Fin ks#    → Un n
  u1     :             Un n
  _ ⊕ _  : (ty tv : Un n) → Un n
  _ ⊗ _  : (ty tv : Un n) → Un n
```

Constructor **I** refers to the  $n$ -th type variable whereas **K** refers to a constant type. Value  $ks\#$  is passed as a module parameter. The denotation is defined as:

```
Parms : ℕ → Set1
Parms n = Fin n → Set

[[_]] : {n : ℕ} → Un n → Parms n → Set
[[ I x   ]] A = A x
[[ K x   ]] A = lookup x ks
[[ u1    ]] A = Unit
[[ ty ⊕ tv ]] A = [[ ty ]] A ⊔ [[ tv ]] A
[[ ty ⊗ tv ]] A = [[ ty ]] A × [[ tv ]] A
```

Note that here, **Parms**  $n$  really is isomorphic to  $n$  types that serve as the parameters to the functor  $[[F]]$ . When we introduce a fixpoint combinator, these parameters are used to tie the recursion knot, just like a simple fixpoint:  $\mu F \equiv F(\mu F)$ . In fact, a mutually recursive family can be easily encoded in this setting. All we need is  $n$  types that refer to  $n$  type-variables each!

```
Fam : ℕ → Set
Fam n = Vec (Un n) n
```

```
data Fix {n : ℕ} (F : Fam n) : Fin n → Set where
  (⟦_⟧) : ∀ {k} → [lookup k F] (Fix F) → Fix F k
```

This universe is enough to model Context-Free grammars, and hence, provides the basic bare bones for diffing elements of an arbitrary programming language. In the future, it could be interesting to see what kind of diffing functionality indexed functors could provide, as these could have scoping rules and other advanced features built into them.

### 1.1 Agda Details

Here we clarify some Agda specific details that are agnostic to the big picture. This can be safely skipped on a first iteration.

As we mentioned above, our codes represent functors on  $n$  variables. Obviously, to program with them, we need to apply these to something. The denotation receives a function  $\text{Fin } n \rightarrow \text{Set}$ , denoted  $\text{Params } n$ , which can be seen as a valuation for each type variable.

In the following sections, we will be dealing with values of  $\llbracket ty \rrbracket_A$  for some class of valuations  $A$ , though. We need to have decidable equality for  $A \ k$  and some mapping from  $A \ k$  to  $\mathbb{N}$  for all  $k$ . We call such valuations a *well-behaved parameter*:

```
record WBParms {n : ℕ} (A : Params n) : Set where
  constructor wb-params
  field
    parm-size : ∀ {k} → A k → ℕ
    parm-cmp   : ∀ {k} (x y : A k) → Dec (x ≡ y)
```

I still have no good justification for the *parm-size* field. Later on I sketch what I believe is the real meaning of the cost function.

The following sections discuss functionality that does not depend on *parameters to codes*. We will be passing them as Agda module parameters. The first diffing technique we discuss is the trivial diff. It's module is declared as follows:

```
module RegDiff.Diff.Trivial.Base
  {ks# : ℕ} (ks : Vec Set ks#) (keqs : Vec1 Eq ks)
  {parms# : ℕ} (A : Params parms#) (WBA : WBParms A)
  where
```

We stick to this nomenclature throughout the code. The first line handles constant types:  $ks\#$  is how many constant types we have,  $ks$  is the vector of such types and  $keqs$  is an indexed vector with a proof of decidable equality over such types. The second line handles type parameters:  $parms\#$  is how many type-variables our codes will have,  $A$  is the valuation we are using and  $WBA$  is a proof that  $A$  is *well behaved*.

We then declare the following synonyms:

```
U : Set
U = Un parms#

sized : {p : Fin parms#} → A p → ℕ
sized = parm-size WBA

_≡-A_ : {p : Fin parms#} (x y : A p) → Dec (x ≡ y)
_≡-A_ = parm-cmp WBA

UUSet : Set1
UUSet = U → U → Set
```

## 2 Computing and Representing Patches

Intuitively, a *Patch* is some description of a transformation. Setting the stage, let  $A$  and  $B$  be a types,  $x : A$  and  $y : B$  elements of such types. A *patch* between  $x$  and  $y$  must specify a few parts:

- i) An  $\text{apply}_p : A \rightarrow \text{Maybe } B$  function,
- ii) such that  $\text{apply}_p x \equiv \text{just } y$ .

Well,  $\text{apply}_p$  can be seen as a functional relation ( $R$  is functional iff  $\text{img } R \subseteq \text{id}$ ) from  $A$  to  $B$ . We call this the “application” relation of the patch, and we will denote it by  $p^\flat \subseteq A \times B$ .

### Needs discussion:

There is still a lot that could be said about this. I feel like  $p^b$  should also be invertible in the sense that:

- i) Let  $(\text{inv } p)$  denote the inverse patch of  $p$ , which is a patch from  $B$  to  $A$ .
- ii) Then,  $p^b \cdot (\text{inv } p)^b \subseteq \text{id}$  and  $(\text{inv } p)^b \cdot p^b \subseteq \text{id}$ . Assuming  $(\text{inv } p)^b$  is also functional, we can use the maybe monad to represent these relations in **Set**. Writing the first equation on a diagram in **Set**, using the *apply* functions:

$$\begin{array}{ccccc}
 B & \xrightarrow{\text{apply}_{(\text{inv } p)}} & A + 1 & \xrightarrow{\text{apply}_p + \text{id}} & (B + 1) + 1 \\
 & \searrow \iota_1 & & & \downarrow \mu \\
 & & & & B + 1
 \end{array}$$

- iii) This is hard to play ball with. We want to say, in a way, that  $x (p^b) y$  iff  $y ((\text{inv } p)^b) x$ . That is,  $(\text{inv } p)$  is the actual inverse of  $p$ . Using relations, one could then say that  $(\text{inv } p)^b$  is the converse of  $(p^b)$ . That is:  $(\text{inv } p)^b \equiv (p^b)^\circ$ . But, if  $(\text{inv } p)^b$  is functional, so is  $(p^b)^\circ$ . This is the same as saying that  $p^b$  is entire! If  $p^b$  is functional and entire, it is a function (and hence, total!). And that is not true.

Now, let us discuss some code and build some intuition for what is what in the above schema. We will present different parts of the code, how do they relate to this relational view and give examples here and there!

## 2.1 Trivial Diff

The simplest possible way to describe a transformation is to say what is the source and what is the destination of such transformation. This can be accomplished by the Diagonal functor just fine.

$\Delta : \mathbf{UUSet}$

$$\Delta \text{ ty } tv = \llbracket \text{ty} \rrbracket A \times \llbracket \text{tv} \rrbracket A$$

Now, take an element  $(x, y) : \Delta \text{ ty } tv$ . The “application” relation it defines is trivial:  $\{(x, y)\}$ , or, in PF style:

$$\begin{array}{ccc}
 & \xrightarrow{\underline{y} \cdot \underline{x}^\circ} & \\
 \llbracket \text{ty} \rrbracket_A & \xleftarrow{\underline{x}} K \xrightarrow{\underline{y}} & \llbracket \text{tv} \rrbracket_A
 \end{array}$$

Where, for any  $A, B \in \mathbf{Set}$  and  $x : A$ ,  $\underline{x} \subseteq A \times B$  represents the *everywhere*  $x$  relation, defined by

$$\underline{x} = \{(x, b) \mid b \in B\}$$

This is a horrible patch however: We can’t calculate with it because we don’t know *anything* about *how*  $x$  changed into  $y$ . Note, however, that  $(x, y)^b \equiv \underline{y} \cdot \underline{x}^\circ$  is trivially functional.

### Needs discussion:

In the code, we actually define the “application” relation of  $\Delta$  as:

$$(x, x)^b = id$$

$$(x, y)^b = \underline{y} \cdot \underline{x}^\circ$$

This suggests that copies might be better off being handled by the trivial diff. We will return to this discussion in section 3

## 2.2 Spines

We can try to make it better by identifying the longest prefix of constructors where  $x$  and  $y$  agree, before giving up and using  $\Delta$ . Moreover, this becomes much easier if  $x$  and  $y$  actually have the same type. In practice, we are only interested in diffing elements of the same language. It does not make sense to diff a C source file against a Haskell source file.

Nevertheless, we define an  $S$  structure to capture the longest common prefix of  $x$  and  $y$ :

```
data S (P : UUSet) : U → Set where
  SX  : {ty : U} → P ty ty → S P ty
  Scp : {ty : U} → S P ty
  S⊗  : {ty tv : U}
    → S P ty → S P tv → S P (ty ⊗ tv)
  Si1 : {ty tv : U}
    → S P ty → S P (ty ⊕ tv)
  Si2 : {ty tv : U}
    → S P ty → S P (tv ⊕ ty)
```

Note that  $S$  makes a functor (actually, a free monad!) on  $P$ , and hence, we can map over it:

```
S-map : {ty : U} {P Q : UUSet}
  → (f : ∀ {k} → P k k → Q k k)
  → S P ty → S Q ty
S-map f (SX x) = SX (f x)
S-map f Scp    = Scp
S-map f (S⊗ s o) = S⊗ (S-map f s) (S-map f o)
S-map f (Si1 s) = Si1 (S-map f s)
S-map f (Si2 s) = Si2 (S-map f s)
```

Computing a spine is easy, first we check whether or not  $x$  and  $y$  are equal. If they are, we are done. If not, we inspect the first constructor and traverse it. Then we repeat.

```
mutual
  spine-cp : {ty : U} → [ ty ] A → [ ty ] A → S Δ ty
  spine-cp {ty} x y
    with dec-eq  $\underline{\quad} \stackrel{?}{=} A \underline{\quad}$  ty x y
  ...| no  _ = spine x y
  ...| yes _ = Scp

  spine : {ty : U} → [ ty ] A → [ ty ] A → S Δ ty
  spine {ty ⊗ tv} (x1 , x2) (y1 , y2)
    = S⊗ (spine-cp x1 y1) (spine-cp x2 y2)
  spine {tv ⊕ tw} (i1 x)   (i1 y) = Si1 (spine-cp x y)
  spine {tv ⊕ tw} (i2 x)   (i2 y) = Si2 (spine-cp x y)
  spine {ty}      x       y     = SX (delta {ty} {ty} x y)
```

The “application” relations specified by a spine  $s = \text{spine-cp } x \ y$ , denoted  $s^b$  are:

$$\begin{aligned}
\text{Scp}^b &= A \xleftarrow{id} A \\
(\text{S} \otimes s_1 \ s_2)^b &= A \times B \xleftarrow{s_1^b \times s_2^b} A \times B \\
(\text{Si1 } s)^b &= A + B \xleftarrow{\iota_1 \cdot s^b \cdot \iota_1^\circ} A + B \\
(\text{Si2 } s)^b &= A + B \xleftarrow{\iota_2 \cdot s^b \cdot \iota_2^\circ} A + B \\
(\text{SX } p)^b &= A \xleftarrow{p^b} A
\end{aligned}$$

Note that, in the  $(\text{SX } p)$  case, we simply ask for the “application” relation of  $p$ .

#### Needs discussion:

Non-canoncity can be a problem:  $\text{Scp}^b \equiv (\text{S} \otimes \text{Scp } \text{Scp})^b$  Even though the `spine-cp` function will never find the right-hand above, it feels sub-optimal to allow this.

One possible solution could be to remove `Scp` and handle them through the maybe monad. Instead of  $\text{S } \Delta$  we would have  $\text{S } (\text{Maybe } \Delta)$ , where the `nothings` represent copy. This ensures that we can only copy on the leaves. Branch `explicit-cpy` of the repo has this experiment going. It is easier said than done.

Ignoring the problems and moving forward; note that for any  $x$  and  $y$ , a spine  $s = \text{spine-cp } x \ y$  will NEVER contain a product nor a unit on a leaf (we force going through products and copying units). Hence, whenever we are traversing  $s$  and find a `SX`, we know that: (1) the values of the pair are different and (2) we must be at a coproduct, a constant type or a type variable. The constant type or the type variable are out of our control. But we can refine our *description* in case we arrive at a coproduct.

## 2.3 Coproduct Changes

Let’s imagine we are diffing the following values of a type  $\mathbb{1} + (\mathbb{N}^2 + \mathbb{N}) \times \mathbb{N}$ :

$$\begin{aligned}
s &= \text{spine-cp } (\iota_2 (\iota_1 (4, 10), 5)) (\iota_2 (\iota_2 10, 5)) \\
&= \text{Si2 } (\text{S} \otimes (\text{SX } (\iota_1 (4, 10), \iota_2 10)) \text{Scp})
\end{aligned}$$

And now, we want to `S-map` over  $s$  and refine the  $\Delta$  inside. We want to have information about which injections we need to pattern match on and which we need to introduce. We begin with a type (that’s also a free monad) that encodes the injections we need to insert or pattern-match on:

```

data C (P : UUSet) : U → U → Set where
  CX  : {ty tv : U} → P ty tv → C P ty tv
  Ci1 : {ty tv k : U} → C P ty tv → C P ty (tv ⊕ k)
  Ci2 : {ty tv k : U} → C P ty tv → C P ty (k ⊕ tv)
  Ci1° : {ty tv k : U} → C P ty tv → C P (ty ⊕ k) tv
  Ci2° : {ty tv k : U} → C P ty tv → C P (k ⊕ ty) tv

```

Note that `C` is also a functor on  $P$ :

```

C-map : {ty tv : U} {P Q : UUSet}
  → (f : ∀ {k v} → P k v → Q k v)
  → C P ty tv → C Q ty tv
C-map f (CX x) = CX (f x)
C-map f (Ci1 s) = Ci1 (C-map f s)
C-map f (Ci2 s) = Ci2 (C-map f s)
C-map f (Ci1° s) = Ci1° (C-map f s)
C-map f (Ci2° s) = Ci2° (C-map f s)

```

We just need an eager function that will consume ALL coproduct injections from both source and destination and record them using  $\mathbf{C}$ :

```

change : {ty tv : U} → [ ty ] A → [ tv ] A → C Δ ty tv
change {ty} {tv ⊕ tw} x (i1 y) = Ci1 (change x y)
change {ty} {tv ⊕ tw} x (i2 y) = Ci2 (change x y)
change {ty ⊕ tw} {tv} (i1 x) y = Ci1° (change x y)
change {ty ⊕ tw} {tv} (i2 x) y = Ci2° (change x y)
change {ty} {tv} x y = CX (delta {ty} {tv} x y)

```

Now, we could  $\mathbf{S}$ -map  $\mathbf{change}$  over  $s$ :

$$\begin{aligned} \mathbf{S}\text{-map } \mathbf{change} \ s &= \mathbf{Si2} (\mathbf{S} \otimes (\mathbf{SX} (\mathbf{Ci2} (\mathbf{Ci1}^\circ (\mathbf{CX} ((\mathcal{A}, 10), 10)))) \mathbf{Scp}) \\ &= s' \end{aligned}$$

The whole thing also becomes of a much more expressive type:

$$s' = \mathbf{S}\text{-map } \mathbf{change} \ s : \mathbf{S} (\mathbf{C} \Delta)$$

We can read the type as: a common prefix from both terms followed by injections into the target term and pattern matching on the source term followed by point-wise changes from source to destination.

Note that we can also say, for sure, that we will never have a  $\mathbf{SX} (\mathbf{Ci1} (\mathbf{Ci1}^\circ c))$  at the  $\mathbf{S}$ -leaves of  $s'$ ; this would mean that we need to inject the target with  $\iota_1$  and pattern match the source on  $\iota_1$ , but this is a common-prefix! Hence it would be recognized by  $\mathbf{spine}\text{-}\mathbf{cp}$  and instead we would have:  $\mathbf{Si1} (\mathbf{SX} c)$ .

#### Needs discussion:

Even though the algorithm does not produce  $\mathbf{SX} (\mathbf{Ci1} (\mathbf{Ci1}^\circ c))$ , as mentioned above, this is still a valid inhabitant of  $\mathbf{S} (\mathbf{C} \Delta)$ !

Just like the values of  $\mathbf{S}$ , we can also define the “application” relation induced by the values of  $\mathbf{C}$ :

$$\begin{aligned} (\mathbf{Ci1} \ c)^b &= B + X \xleftarrow{\iota_1 \cdot c^b} A \\ (\mathbf{Ci2} \ c)^b &= X + B \xleftarrow{\iota_2 \cdot c^b} A \\ (\mathbf{Ci1}^\circ \ c)^b &= B \xleftarrow{c^b \cdot \iota_1^\circ} A + X \\ (\mathbf{Ci2}^\circ \ c)^b &= B \xleftarrow{c^b \cdot \iota_2^\circ} X + A \\ (\mathbf{CX} \ p)^b &= B \xleftarrow{p^b} A \end{aligned}$$

Note that up until now, everything was deterministic! This is something we are about to lose.

## 2.4 Aligning Everything

Following a similar reasoning as from  $\mathbf{S}$  to  $\mathbf{C}$ ; the leaves of a  $\mathbf{C}$  produced through  $\mathbf{change}$  will NEVER contain a coproduct as the topmost type. Hence, we know that they will contain either a product, or a constant type, or a type variable. In the case of a constant type or a type variable, there is not much we can do at the moment, but for a product we can refine this a little bit more before using  $\Delta^1$ .

Looking at our running example, we have a leaf  $(\mathbf{CX} ((\mathcal{A}, 10), 10))$  to take care of. Here the source type is  $\mathbb{N}^2$  and the destination is  $\mathbb{N}$ . Note that the  $\mathbb{N}$  is treated as a constant type here. As we mentioned above, we have a product on the source, so we could extract some more information before giving up and using  $\Delta$ !

<sup>1</sup>In fact, splitting the different stages of the algorithm into different types reinforced our intuition that the alignment is the source of difficulties. As we shall see, we now need to introduce non-determinism.

### 2.4.1 A Parenthesis

On the literature for version control system, the *alignment* problem is the problem of mapping two strings  $l_1$  and  $l_2$  in  $\mathcal{L}$  into  $\mathcal{L} \cup \{-\}$ , for  $\{-\} \not\subseteq \mathcal{L}$  such that the resulting strings  $l'_1$  and  $l'_2$  are of the same length such that for all  $i$ , it must not be happen that  $l'_1[i] = - = l'_2[i]$ . For example, Take strings  $l_1 = \text{"CGTCG"}$  and  $l_2 = \text{"GATAGT"}$ , then, the following is an (optimal) alignment:

C	G	-	T	C	G	-
-	G	A	T	A	G	T

Let  $\mathcal{DNA} = \{A, T, C, G\}$ . Finding the table above is the same as finding a partial map:

$$f : \mathcal{DNA}^5 \rightarrow \mathcal{DNA}^6$$

such that  $f(C, G, T, C, G) = (G, A, T, A, G, T)$ . There are many ways of defining such a map. We would like, however, that our definition have a maximal domain, that is, we require impose the least possible amount of restrictions. In this case, we can actually define  $a$  with some pattern matching as:

$$\begin{aligned} f(C, x, y, C, z) &= (x, A, y, A, z, T) \\ f \quad - &= \text{undefined} \end{aligned}$$

And it is easy to verify that, in fact,  $f(C, G, T, C, G) = (G, A, T, A, G, T)$ . Moreover, this is the *maximal* such  $f$  that still (provably) assigns the correct destination to the correct source.

### 2.4.2 Back to Agda

We will look at alignments from the “finding a map between products” perspective. Here is where our design space starts to be huge, and so, we should start making some distinctios:

- We want to allow sharing. This means that there can be more than one variable in the defining pattern of our  $f$ .
- We do *not* allow permutations, as the search space would be too big. This means that the variables appear in the right-hand side of  $f$  in the same order as they appear in the left-hand-side.
- We do *not* allow contractions nor weakenings. That is, every variable on the left-hand-side of  $f$  must appear *exactly* once on the right-hand-side.

The following datatype describe such maps:

```
data AI (P : UUSet) : U → U → Set where
  AX  : {ty tv : U}      → P ty tv → AI P ty tv
  Ap1 : {ty tv tw : U}   → [ tw ] A → AI P ty tv → AI P ty (tv ⊗ tw)
  Ap1° : {ty tv tw : U}  → [ tw ] A → AI P ty tv → AI P (ty ⊗ tw) tv
  Ap2 : {ty tv tw : U}   → [ tw ] A → AI P ty tv → AI P ty (tw ⊗ tv)
  Ap2° : {ty tv tw : U}  → [ tw ] A → AI P ty tv → AI P (tw ⊗ ty) tv
  A⊗  : {ty tv tw tz : U} → AI P ty tw → AI P tv tz → AI P (ty ⊗ tv) (tw ⊗ tz)
```

Computing alignments is very expensive, specially if done in the naive way. Nevertheless, we present the naive alignment first, and distinguish it with an *exp* suffix. In the case we actually have products on both the source and the destination we have a lot of options (hence the list monad!):

```

align-exp : {ty tv : U} → [ ty ] A → [ tv ] A → List (Al Δ ty tv)
align-exp {ty ⊗ ty} {tv ⊗ tv} (x1 , x2) (y1 , y2)
  = A⊗ <$> align-exp x1 y1 <*> align-exp x2 y2
  ++ Ap1 y2 <$> align-exp (x1 , x2) y1
  ++ Ap2 y1 <$> align-exp (x1 , x2) y2
  ++ Ap1° x2 <$> align-exp x1 (y1 , y2)
  ++ Ap2° x1 <$> align-exp x2 (y1 , y2)

```

If we only have products on the left or on the right (or none), we have less options:

```

align-exp {ty ⊗ ty} {tv} (x1 , x2) y
  = Ap1° x2 <$> align-exp x1 y
  ++ Ap2° x1 <$> align-exp x2 y
align-exp {ty} {tv ⊗ tv} x (y1 , y2)
  = Ap1 y2 <$> align-exp x y1
  ++ Ap2 y1 <$> align-exp x y2
align-exp {ty} {tv} x y = return (AX (x , y))

```

Let's come back to our running example! Following the previous trend, we could **C-mapM** our alignment function (the **C-mapM** is the monadic variant of **C-map**) on  $s'$ , in order to find all alignments of (4, 10) and 10. In this case, this is very easy<sup>2</sup>.

```

C-map align s' = [ Si2 (S⊗ (SX (Ci2 (Ci1° (CX (Ap1° 10 (AX (4 , 10))))))) Scp)
  , Si2 (S⊗ (SX (Ci2 (Ci1° (CX (Ap2° 4 (AX (10 , 10))))))) Scp ) ]

```

If we omit the **SX** and **CX** constructors this becomes slightly more readable:

```

C-map align s' = [ Si2 (S⊗ (Ci2 (Ci1° (Ap1° 10 (AX (4 , 10)))))) Scp)
  , Si2 (S⊗ (Ci2 (Ci1° (Ap2° 4 (AX (10 , 10)))))) Scp ] ]

```

But now we end up having to choose between one of those to be *the* patch. This is where we start to need a cost function. Before talking about cost, let's look at the “application” relations of **Al**:

$$\begin{aligned}
(\mathbf{A}\otimes a_1 a_2)^b &= B \times D \xleftarrow{a_1^b \times a_2^b} A \times C \\
(\mathbf{Ap1} x a)^b &= B \times X \xleftarrow{\langle a^b, \underline{x} \rangle} A \\
(\mathbf{Ap2} x a)^b &= X \times B \xleftarrow{\langle \underline{x}, a^b \rangle} A \\
(\mathbf{Ap1}^\circ x a)^b &= B \xleftarrow{\pi_1 \cdot (a^b \times \underline{x}^\circ)} A \times X \\
(\mathbf{Ap2}^\circ x a)^b &= B \xleftarrow{\pi_2 \cdot (\underline{x}^\circ \times a^b)} X \times A \\
(\mathbf{AX} p)^b &= B \xleftarrow{p^b} A
\end{aligned}$$

Note how **Ap1°** and **Ap2°** force a component of the source to be equal to something (that's equivalent to pattern matching on the source) whereas **Ap1** and **Ap2** set a component of the destination to be equal to something. Let us represent the  $f$  we devised on the  $\mathcal{DNA}$  example using **Al**. Recall  $f(C, x, y, C, z) = (x, A, y, A, z, T)$ .

```

f ≡ Ap2° C (A⊗ Scp (Ap2 A (A⊗ Scp (A⊗ (AX (C , A)) (Ap1 T Scp)))))

```

---

<sup>2</sup>It might be hard to build intuition for why we need the **A⊗** constructor. On the **Lab** module of **Fixpoint** there is an example using 2-3-Trees that motivates the importance of that constructor



## TODO

Yet... for some reason, our simple optimization below is not finding the above alignment... we need to fix it!

I can see Tarmo's coalgebras for sharing working wonders in computing these alignments in linear time.

## 2.5 A simple optimization

### Needs discussion:

Looking carefully at the `align-exp` function above, we are computing a lot of unnecessary branches, specially in the case for products on both sides. Below we compute `align-exp (x , y) (w , z)` and put arrows indicating which `Al` have an isomorphic underlying “application” relation.

$$\begin{aligned}
 \text{align-exp } (x, y) (w, z) &= A \otimes (AX(x, w)) (AX(y, z)) \\
 &\quad \begin{aligned}
 &:: \text{Ap1 } z (\text{Ap1}^\circ y (AX(x, w))) \\
 &:: \text{Ap1 } z (\text{Ap2}^\circ x (AX(y, w))) \\
 &:: \text{Ap2 } w (\text{Ap1}^\circ y (AX(x, z))) \\
 &:: \text{Ap2 } w (\text{Ap2}^\circ x (AX(y, z))) \\
 &:: \text{Ap1}^\circ y (\text{Ap1 } z (AX(x, w))) \\
 &:: \text{Ap2}^\circ x (\text{Ap1 } z (AX(y, w))) \\
 &:: \text{Ap1}^\circ y (\text{Ap2 } w (AX(x, z))) \\
 &:: \text{Ap2}^\circ x (\text{Ap2 } w (AX(y, z))) \\
 &:: []
 \end{aligned}
 \end{aligned}$$

If we then look at which `Alignments` do *not* have an arrow out of it, that is, the ones that don't have an isomorphic alignment also being computed, we get:

$$\begin{aligned}
 \text{align } (x, y) (w, z) &= A \otimes (AX(x, w)) (AX(y, z)) \\
 &\quad \begin{aligned}
 &:: \text{Ap1 } z (\text{Ap2}^\circ x (AX(y, w))) \\
 &:: \text{Ap2 } w (\text{Ap1}^\circ y (AX(x, z))) \\
 &:: []
 \end{aligned}
 \end{aligned}$$

### Needs discussion:

We hence simplify the alignment function:

```

align : {ty tv : U} → [ ty ] A → [ tv ] A → List (A Δ ty tv)
align {ty ⊗ ty} {tv ⊗ tv} (x1 , x2) (y1 , y2)
  = A ⊗ <$> align x1 y1 <*> align x2 y2
  ++ (Ap1 y2 ∘ Ap2° x1) <$> kill-p2 (align x2 y1)
  ++ (Ap2 y1 ∘ Ap1° x2) <$> kill-p1 (align x1 y2)

align {ty ⊗ ty} {tv} (x1 , x2) y
  = Ap1° x2 <$> align x1 y
  ++ Ap2° x1 <$> align x2 y
align {ty} {tv ⊗ tv} x (y1 , y2)
  = Ap1 y2 <$> align x y1
  ++ Ap2 y1 <$> align x y2
align {ty} {tv} x y = return (AX (x , y))

```

The `kill-p2` (resp. `kill-p1`) functions above are used to prune the search space further. They will ignore every branch that starts with a `Ap2` (resp `Ap1`), which will, for sure, have had an isomorphic alignment computed already (in terms of `A ⊗`). Writing down the “application” relations makes this easy to see.


## 3 Patches as Relations

Here we will be using the semantics for  $(x , y)^b$  as described in the discussion box at Section 2.1. That is, we would like to say that the patch (P2), above, copies the 10, instead of saying that 10 changes into 10. We can postpone this by changing the relation semantics of  $\Delta$ . Hence:  $(x, x)^{b_i} = id$  and  $(x, y)^b = \underline{y} \cdot \underline{x}^\circ$ .

Nevertheless, if we look at the two patches above as relations, we have:

$$\begin{aligned}
 P1^b &= i_2 \cdot (i_2 \cdot <\underline{4} \cdot \underline{10}^\circ, \underline{10}>^\circ \cdot i_1^\circ \times id) \cdot i_2^\circ \\
 P2^b &= i_2 \cdot (i_2 \cdot <\underline{4}, id>^\circ \cdot i_1^\circ \times id) \cdot i_2^\circ
 \end{aligned}$$

Writing them in a diagram:



Here, we have something curious going on... We have that  $P1^b \subseteq P2^b$ . To see this is not very hard. First, composition and converses are monotonous with respect to  $\subseteq$ . We are left to check that:

$$\begin{aligned}
 &<\underline{4} \cdot \underline{10}^\circ, \underline{10}> \subseteq <\underline{4}, id> \\
 &\equiv \{ \text{split universal} \} \\
 &\pi_1 \cdot <\underline{4} \cdot \underline{10}^\circ, \underline{10}> \subseteq \underline{4} \wedge \pi_2 \cdot <\underline{4} \cdot \underline{10}^\circ, \underline{10}> \subseteq id
 \end{aligned}$$

The first proof obligation is easy to calculate with:

$$\begin{aligned}
& \pi_1 \cdot < \underline{4} \cdot \underline{10}^\circ, \underline{10} > \subseteq \underline{4} \\
& \Leftarrow \{ \pi_1\text{-cancel} ; \subseteq\text{-trans} \} \\
& \underline{4} \cdot \underline{10}^\circ \subseteq \underline{4} \\
& \Leftarrow \{ \text{Leibniz} \} \\
& \underline{4} \cdot \underline{10}^\circ \cdot \underline{10} \subseteq \underline{4} \cdot \underline{10} \\
& \equiv \{ \underline{a}^\circ \cdot \underline{a} \equiv \top \} \\
& \underline{4} \cdot \top \subseteq \underline{4} \cdot \underline{10} \\
& \equiv \{ \underline{a} \cdot \underline{b} \equiv \underline{a} \} \\
& \underline{4} \cdot \top \subseteq \underline{4} \\
& \equiv \{ \underline{a} \cdot \top \equiv \underline{a} \} \\
& \underline{4} \subseteq \underline{4} \\
& \equiv \{ \subseteq\text{-refl} \} \\
& \text{True}
\end{aligned}$$

The second is easier to prove once we add variables!

$$\begin{aligned}
& \pi_2 \cdot < \underline{4} \cdot \underline{10}^\circ, \underline{10} > \subseteq \text{id} \\
& \equiv \{ \text{Add variables} \} \\
& \forall x, y . x (\pi_2 \cdot < \underline{4} \cdot \underline{10}^\circ, \underline{10} >) y \Rightarrow x = y \\
& \equiv \{ \text{PF expand composition} \} \\
& \forall x, y . \exists z . x (\pi_2) z \wedge z < \underline{4} \cdot \underline{10}^\circ, \underline{10} > y \Rightarrow x = y \\
& \equiv \{ \text{Types force } z = (z_1, z_2) \} \\
& \forall x, y . \exists z_1, z_2 . x (\pi_2) (z_1, z_2) \wedge (z_1, z_2) < \underline{4} \cdot \underline{10}^\circ, \underline{10} > y \Rightarrow x = y \\
& \equiv \{ \pi_2 \text{ def} \} \\
& \forall x, y . \exists z_1, z_2 . x = z_2 \wedge (z_1, z_2) < \underline{4} \cdot \underline{10}^\circ, \underline{10} > y \Rightarrow x = y \\
& \equiv \{ \text{split def} \} \\
& \forall x, y . \exists z_1, z_2 . x = z_2 \wedge z_1 (\underline{4} \cdot \underline{10}^\circ) y \wedge z_2 (\underline{10}) y \Rightarrow x = y \\
& \equiv \{ \text{points def} \} \\
& \forall x, y . \exists z_1, z_2 . x = z_2 \wedge z_1 = 4 \wedge y = 10 \wedge z_2 = 10 \Rightarrow x = y \\
& \equiv \{ \text{substitutions ; weakening} \} \\
& \forall x, y . \exists z_2 . x = 10 \wedge y = 10 \Rightarrow x = y \\
& \equiv \{ \text{trivial} \} \\
& \text{True}
\end{aligned}$$

Nevertheless, it is clear which patch we should choose! We should always choose the patch that gives rise to the biggest relation, as this is applicable to much more elements.

This suggests an interesting justification for the cost function. For some reason, looks like we won the lottery with our cost functions. We are always choosing the patch that gives rise to the maximal relation. I still don't clearly understand why or how, but it works.

## TODO

Below are our cost functions:

we should actually be able to choose patches without cost functions... discuss this up! Use the [align](#) optimization example...

ADD VALUES, FORGET ABOUT THOSE RAW FUNCTIONS!

```

cost-Δ : {ty tv : U} → Δ ty tv → ℕ
cost-Δ {ty} {tv} (x , y) with U-eq ty tv
cost-Δ {ty} {ty} (x , y) | yes refl
    with dec-eq _?=-A_ ty x y
...| yes _ = 0
...| no _ = cost-Δ-raw {ty} {ty} (x , y)
cost-Δ {ty} {tv} (x , y) | no _
    = cost-Δ-raw {ty} {tv} (x , y)

S-cost : {ty : U}{P : UUSet}
    → (costP : ∀{ty} → P ty ty → ℕ)
    → S P ty → ℕ
S-cost c (SX x) = c x
S-cost c Scp = 0
S-cost c (S⊗ s o) = S-cost c s + S-cost c o
S-cost c (Si1 s) = S-cost c s
S-cost c (Si2 s) = S-cost c s

C-cost : {ty tv : U}{P : UUSet}
    → (costP : ∀{ty tv} → P ty tv → ℕ)
    → C P ty tv → ℕ
C-cost c (CX x) = c x
C-cost c (Ci1 s) = C-cost c s
C-cost c (Ci2 s) = C-cost c s
C-cost c (Ci1° s) = C-cost c s
C-cost c (Ci2° s) = C-cost c s

Al-cost : {ty tv : U}{P : UUSet}
    → (costP : ∀{ty tv} → P ty tv → ℕ)
    → Al P ty tv → ℕ
Al-cost c (AX xy) = c xy
Al-cost c (A⊗ s o) = Al-cost c s + Al-cost c o
Al-cost c (Ap1 {tw = k} x s) = Al-cost-raw {k} x (Al-cost c s)
Al-cost c (Ap2 {tw = k} x s) = Al-cost-raw {k} x (Al-cost c s)
Al-cost c (Ap1° {tw = k} x s) = Al-cost-raw {k} x (Al-cost c s)
Al-cost c (Ap2° {tw = k} x s) = Al-cost-raw {k} x (Al-cost c s)

```

### 3.1 Patches for Regular Types

Now that we have *spines*, *changes* and *alignments* figured out, we can define a patch as:

```

Patch : U → Set
Patch ty = S (C (Al Δ)) ty

```

Computing inhabitants of such type is done with:

```

diff1* : {ty : U} → [ ty ] A → [ ty ] A → Patch* ty
diff1* x y = S-mapM (C-mapM (uncurry align) ∘ uncurry change) (spine-cp x y)

```

## 4 Mutually Recursive Types

Now that we have a clear picture of regular types, extending this to recursive types is not very difficult.

First, recall that a mutually recursive family is defined as  $n$  codes that each reference  $n$  type variables:

```
Fam : ℕ → Set
Fam n = Vec (Un n) n

data Fix {n : ℕ} (F : Fam n) : Fin n → Set where
  ⟨_⟩ : ∀ {k} → [ lookup k F ] (Fix F) → Fix F k
```

Another auxiliary definition we use here is the indexed coproduct, which let's us *extend* some indexed type.

```
+u : {n : ℕ} → (Un n → Un n → Set) → (Un n → Un n → Set) → (Un n → Un n → Set)
(P +u Q) ty tv = (P ty tv) ⊔ (Q ty tv)
```

Now, we already have the ingredients for common prefixes, coproducts and products. We now need to handle type variables. Before we proceed with the nasty definitions, we still need two last synonyms:

```
Fami : Set
Fami = Fin fam#

T : Fami → Un fam#
T k = lookup k fam
```

Here  $T\ k$  represents the  $k$ -th type of the family, and  $Fam_i$  represents the indexes of our family.

Remember that we have **Spines**, to handle the common prefixes; **Changes** to handle different coproduct injections on source and destination and **Alignments** to make sure we exploit product structure. We kept mentioning that we could not do anything on the leaves that were type variables or constant types. Well, constant types we still can't, and we have to stick to  $\Delta$ , but now we need to handle the type-variables since they are used to tie the recursion knot.

A patch for a fixed point might not follow the precise order of operations (**S**, then **C**, then **Al**) that regular types enjoyed. For instance, imagine we are transforming the following lists:

$$[5, 8, 13, 21] \rightsquigarrow [8, 13, 21]$$

Let lists be seen (as usual) as the initial algebras of  $L_A\ X = 1 + A \times X$ ; then, both lists are inhabitants of  $\mu L_N$ , but, more precisely, the source is an inhabitant of  $L_N(L_N(L_N(L_N\ 1)))$  whereas the target is an inhabitant of  $L_N(L_N(L_N\ 1))$ .

Here, we are already beginning with different types, so a spine (which is homogeneous) might not be the best start! In fact, the best start is to say that the first 5 is deleted, then the spine can kick in and say that everything else is copied!

Here is the definition:

```
data Patchμ : U → U → Set where
  chng : {ty tv : U} → Cμ Patchμ ty tv → Patchμ ty tv
  fix  : {k k' : Fami} → Patchμ (T k) (T k') → Patchμ (I k) (I k')
  set  : {ty : U} → Δ ty ty → Patchμ ty ty
```

The **fix** constructor ties the know between type-variables and their lookup in the recursive family, which is isomorphic. The **set** constructor specifies that we are setting something. Note that we require the source and destination types to be equal here!

### Needs discussion:

Experimenting with this, I found out that **set** is only used for constant types (as expected!). A proof of this would be great.

$C\mu$  type extends the previous  $C$  with the usual edit operations everyone talks about: modify, delete or insert!

```
data Cμ (P : UUSet) : U → U → Set where
  Cins  : {k k' : Fami} → C (Al P) (l k) (T k') → Cμ P (T k) (T k')
  Cdel  : {k k' : Fami} → C (Al P) (T k) (l k') → Cμ P (T k) (T k')
  Cmod  : {ty : U} → S (C (Al P)) ty → Cμ P ty ty
```

Note how insertions (resp. deletions) happen by *changing and aligning* a type-variable into a type (resp. a type into a type-variable). Note also that `fix`, `Cins` and `Cdel` are heterogeneous on the `Fami` indexes! This is very important for mutually recursive families.

The modifications, therefore, happen when we can take a spine before changing and aligning something.

Computing a `Patchμ` is done by piggybacking on the functions for computing `S`, `C` and `Al` separately, then mapping over them with some refinement functions:

```
changeμ : {ty tv : U}
  → [ ty ] (Fix fam) → [ tv ] (Fix fam)
  → List (C (Al Patchμ) ty tv)
changeμ x y = C-mapM ((_)>= Al-mapM refine-Al) ∘ uncurry align (change x y)

try-mod : {ty tv : U} → [ ty ] (Fix fam) → [ tv ] (Fix fam) → List (Patchμ ty tv)
try-mod {ty} {tv} x y with U-eq ty tv
try-mod {ty} {tv} x y | no _ = []
try-mod {ty} {tv} x y | yes refl
  = (chng ∘ Cmod) <$> S-mapM (uncurry changeμ) (spine-cp x y)

{-# TERMINATING #-}
diffμ* : {k k' : Fami} → Fix fam k → Fix fam k' → List (Patchμ (T k) (T k'))
diffμ* {k} {k'} x y
  = try-mod {T k} {T k'} x y
  ++ ((chng ∘ Cdel {k = k} {k'}) <$> changeμ x y)
  ++ ((chng ∘ Cins {k = k} {k'}) <$> changeμ x y)
```

The refinement function is easy. We just need to make sure we have equal source and destination types before we `set` something; otherwise we simply prune that branch out. This makes the code much more efficient as we are not trying to `set` an integer into a list.

```
mutual
  refine-Al : {k v : U} → Δ k v → List (Patchμ k v)
  refine-Al {l k} {l k'} (x , y) = fix <$> diffμ* x y
  refine-Al {k} {v} (x , y) with U-eq k v
  refine-Al {k} {v} (x , y) | no _ = []
  refine-Al {k} {k} (x , y) | yes refl = return (set (x , y))
```

Obviously, we can also define the “application” relation for fixed points, and that is done in `RegDiff/Diff/Multirec/Domain`. I believe it is more worthwhile to look at some example patches and their “application” relation instead of the general case, though. Let’s begin with the lists we just discussed:

#### TODO

recompile the examples