

Diffing Mutually Recursive Types

A code tour

Victor Cacciari Miraldo

University of Utrecht

December 9, 2016

1 Our Universe

The universe we are using is a *Sums-of-Products* over type variables and constant types.

```
data Atom (n : ℕ) : Set where
  l : Fin n      → Atom n
  K : Fin ks#    → Atom n
```

Constructor `l` refers to the n -th type variable whereas `K` refers to a constant type. Value `ks#` is passed as a module parameter. We denote products by π and sums by σ , but they are just lists.

```
 $\pi$  : ℕ → Set
 $\pi$  = List ∘ Atom
```

```
 $\sigma\pi$  : ℕ → Set
 $\sigma\pi$  = List ∘  $\pi$ 
```

Interpreting these codes is very simple. Here, `Parms` is a valuation for the type variables.

```
Parms : ℕ → Set1
Parm s n = Fin n → Set
```

```
[_]a : {n : ℕ} → Atom n → Parm s n → Set
[l x]a    A = A x
[K x]a    A = lookup x ks
```

```
[_]p : {n : ℕ} →  $\pi$  n → Parm s n → Set
[[]]p    A = Unit
[a :: as]p A = [a]a A × [as]p A
```

```
[_] : {n : ℕ} →  $\sigma\pi$  n → Parm s n → Set
[[]]    A = ⊥
[p :: ps] A = [p]p A ⊔ [ps] A
```

Note that here, `Parms n` really is isomorphic to n types that serve as the parameters to the functor `[F]`. When we introduce a fixpoint combinator, these parameters are used to tie the recursion knot, just like a simple fixpoint: $\mu F \equiv F (\mu F)$. In fact, a mutually recursive family can be easily encoded in this setting. All we need is n types that refer to n type-variables each!

```
Fam : ℕ → Set
Fam n = Vec ( $\sigma\pi$  n) n
```

```
data Fix {n : ℕ} (F : Fam n) : Fin n → Set where
  (<_>) : ∀ {k} → [lookup k F] (Fix F) → Fix F k
```

This universe is enough to model Context-Free grammars, and hence, provides the basic bare bones for diffing elements of an arbitrary programming language. In the

future, it could be interesting to see what kind of diffing functionality indexed functors could provide, as these could have scoping rules and other advanced features built into them.

1.1 SoP peculiarities

One slightly cumbersome problem we have to circumvent is that the codes for type variables and constant types have a different *type* than the codes for types. This requires more discipline to organize our code. Nevertheless, we may wish to see **Atoms** as a trivial *Sum-of-Product*.

```
α : {n : ℕ} → Atom n → σπ n
α a = (a :: []) :: []
```

Instead of having binary injections into coproducts, like we would on a *regular-like* universe, we have n -ary injections, or, *constructors*. We encapsulate the idea of constructors of a $\sigma\pi$ into a type and write a *view* type that allows us to look at an inhabitant of a sum of products as a *constructor* and *data*.

First, we define constructors:

```
cons# : {n : ℕ} → σπ n → ℕ
cons# = length

Constr : {n : ℕ} (ty : σπ n) → Set
Constr ty = Fin (cons# ty)
```

Now, a constructor of type C expects some arguments to be able to make an element of type C . This is a product, we call it the **typeOf** the constructor.

```
typeOf : {n : ℕ} (ty : σπ n) → Constr ty → π n
typeOf [] ()
typeOf (x :: ty) fz = x
typeOf (x :: ty) (fs c) = typeOf ty c
```

Injecting is fairly simple.

```
inject : {n : ℕ} {A : Params n} {ty : σπ n}
  → (i : Constr ty) → [ typeOf ty i ]_p A
  → [ ty ] A
inject {ty = []} () cs
inject {ty = x :: ty} fz cs = i1 cs
inject {ty = x :: ty} (fs i) cs = i2 (inject i cs)
```

We finish off with a *view* of $[ty]_A$ as a constructor and some data. This greatly simplify the algorithms later on.

```
data SOP {n : ℕ} {A : Params n} {ty : σπ n} : [ ty ] A → Set where
  strip : (i : Constr ty) (ls : [ typeOf ty i ]_p A)
    → SOP (inject i ls)
```

1.2 Agda Details

Here we clarify some Agda specific details that are agnostic to the big picture. This can be safely skipped on a first iteration.

As we mentioned above, our codes represent functors on n variables. Obviously, to program with them, we need to apply these to something. The denotation receives a function $\text{Fin } n \rightarrow \text{Set}$, denoted **Params** n , which can be seen as a valuation for each type variable.

In the following sections, we will be dealing with values of $[ty]_A$ for some class of valuations A , though. We need to have decidable equality for A k and some mapping from A k to \mathbb{N} for all k . We call such valuations a *well-behaved parameter*:

```

record WBParms {n : ℕ}(A : Params n) : Set where
  constructor wb-params
  field
    parm-size : ∀{k} → A k → ℕ
    parm-cmp   : ∀{k}(x y : A k) → Dec (x ≡ y)

```

TODO

The field *parm-size* is not really needed anymore! Remove it!

The following sections discuss functionality that does not depend on *parameters to codes*. Hence, we will be passing them as Agda module parameters. We also set up a number of synonyms to already fix the aforementioned parameter. The first diffing technique we discuss is the trivial diff. It's module is declared as follows:

```

module RegDiff.Diff.Trivial.Base
  {ks# : ℕ}(ks : Vec Set ks#)(keqs : Vec1 Eq ks)
  {parms# : ℕ}(A : Params parms#)(WBA : WBParms A)
  where

```

We stick to this nomenclature throughout the code. The first line handles constant types: *ks#* is how many constant types we have, *ks* is the vector of such types and *keqs* is an indexed vector with a proof of decidable equality over such types. The second line handles type parameters: *parms#* is how many type-variables our codes will have, *A* is the valuation we are using and *WBA* is a proof that *A* is *well behaved*.

TODO

Now parameters are setoids, we can drop out the WBA record.

Below are the synonyms we use for the rest of the code:

```

U : Set
U = σπ parms#

Atom : Set
Atom = Atom' parms#

Π : Set
Π = π parms#

sized : {p : Fin parms#} → A p → ℕ
sized = parm-size WBA

_≡?A_ : {p : Fin parms#}(x y : A p) → Dec (x ≡ y)
_≡?A_ = parm-cmp WBA

```

```

[[_]]_a : Atom → Set
[[ a ]]_a = interp_a a A

[[_]]_p : Π → Set
[[ p ]]_p = interp_p p A

[[_]] : U → Set
[[ u ]] = interp_s u A

UUSet : Set1
UUSet = U → U → Set

AASet : Set1
AASet = Atom → Atom → Set

ΠΠISet : Set1
ΠΠISet = Π → Π → Set

contr : ∀{a b}{A : Set a}{B : Set b}
  → (A → A → B) → A → B
contr p x = p x x

UU→AA : UUSet → AASet
UU→AA P a a' = P (α a) (α a')

→α : {a : Atom} → [[ a ]]_a → [[ α a ]]
→α k = il (k, unit)

```

2 Computing and Representing Patches

Intuitively, a *Patch* is some description of a transformation. Setting the stage, let A and B be a types, $x : A$ and $y : B$ elements of such types. A *patch* between x and y must specify a few parts:

- i) An $apply_p : A \rightarrow \text{Maybe } B$ function,
- ii) such that $apply_p x \equiv \text{just } y$.

Well, $apply_p$ can be seen as a functional relation (R is functional iff $\text{img } R \subseteq \text{id}$) from A to B . We call this the “application” relation of the patch, and we will denote it by $p^b \subseteq A \times B$.

Needs discussion:

There is still a lot that could be said about this. I feel like p^b should also be invertible in the sense that:

- i) Let $(\text{inv } p)$ denote the inverse patch of p , which is a patch from B to A .
- ii) Then, $p^b \cdot (\text{inv } p)^b \subseteq \text{id}$ and $(\text{inv } p)^b \cdot p^b \subseteq \text{id}$. Assuming $(\text{inv } p)^b$ is also functional, we can use the maybe monad to represent these relations in **Set**. Writing the first equation on a diagram in **Set**, using the *apply* functions:

$$\begin{array}{ccccc}
 B & \xrightarrow{\text{apply}_{(\text{inv } p)}} & A + 1 & \xrightarrow{\text{apply}_p + \text{id}} & (B + 1) + 1 \\
 & \searrow \iota_1 & & & \downarrow \mu \\
 & & & & B + 1
 \end{array}$$

- iii) This is hard to play ball with. We want to say, in a way, that $x (p^b) y$ iff $y ((\text{inv } p)^b) x$. That is, $(\text{inv } p)$ is the actual inverse of p . Using relations, one could then say that $(\text{inv } p)^b$ is the converse of (p^b) . That is: $(\text{inv } p)^b \equiv (p^b)^\circ$. But, if $(\text{inv } p)^b$ is functional, so is $(p^b)^\circ$. This is the same as saying that p^b is entire! If p^b is functional and entire, it is a function (and hence, total!). And that is not true.

Now, let us discuss some code and build some intuition for what is what in the above schema. We will present different parts of the code, how do they relate to this relational view and give examples here and there!

2.1 Trivial Diff

The simplest possible way to describe a transformation is to say what is the source and what is the destination of such transformation. This can be accomplished by the Diagonal functor, Δ , just fine.

Now, take an element $(x, y) : \Delta \text{ ty } tv$. The “application” relation it defines is trivial: $\{(x, y)\}$, or, in PF style:

$$\begin{array}{ccc}
 & \xrightarrow{\underline{y} \cdot \underline{x}^\circ} & \\
 \llbracket ty \rrbracket_A & \xleftarrow{\underline{x}} K \xrightarrow{\underline{y}} & \llbracket tv \rrbracket_A
 \end{array}$$

Where, for any $A, B \in \text{Set}$ and $x : A$, $\underline{x} \subseteq A \times B$ represents the *everywhere* x relation, defined by

$$\underline{x} = \{(x, b) \mid b \in B\}$$

This is a horrible patch however: We can’t calculate with it because we don’t know *anything* about *how* x changed into y . Note, however, that $(x, y)^b \equiv \underline{y} \cdot \underline{x}^\circ$ is trivially functional.

Needs discussion:

In the code, we actually define the “application” relation of Δ as:

$$\begin{aligned}
 (x, x)^b &= \text{id} \\
 (x, y)^b &= \underline{y} \cdot \underline{x}^\circ
 \end{aligned}$$

This suggests that copies might be better off being handled by the trivial diff. We will return to this discussion in section 3

2.1.1 Trivial Diff, in Agda

We will be using Δ *ty* *tv* for the three levels of our universe: atoms, products and sums. We distinguish between the different Δ 's with subscripts $_a$, $_p$ and $_s$ respectively. They only differ in type. The treatment they receive in the code is exactly the same! Below is how they are defined:

```
delta : ∀{a}{A : Set a}(P : A → Set)
  → A → A → Set
delta P a1 a2 = P a1 × P a2
```

Hence, we define $\Delta_x = \text{delta } \llbracket \cdot \rrbracket_x$, for $x \in \{a, p, s\}$.

2.2 Spines

We can make the trivial diff better by identifying whether or not x and y agree on something! In fact, we will aggressively look for copying opportunities. We start by checking if x and y are, in fact, equal. If they are, we say that the patch that transforms x into y is *copy*. If they are not equal, they might have the same *constructor*. If they do, then say that the constructor is copied and we put the data side by side (zip). Otherwise, there is nothing we can do on this phase and we just return $\Delta x y$.

Note that the *spine* forces x and y to be of the same type! In practice, we are only interested in diffing elements of the same language. It does not make sense to diff a C source file against a Haskell source file.

Nevertheless, we define an **S** structure to capture this longest common prefix of x and y ; which, for the *SoP* universe is very easy to state.

```
data S (P : UUSet) : U → Set where
  SX  : {ty : U} → P ty ty → S P ty
  Scp : {ty : U} → S P ty
  Scns : {ty : U}(i : Constr ty)
    → Listl (contr P ∘ α) (typeOf ty i)
    → S P ty
```

Remember that $\text{contr } P x = P x x$ and $\alpha : \text{Atom } n \rightarrow \sigma\pi n$; Here, $\text{Listl } P l$ is an indexed list where the elements have type $P l_i$, for every $l_i \in l$. We will treat this type like an ordinary list for the remainder of this document.

Note that **S** makes a functor (actually, a free monad!) on P , and hence, we can map over it:

```
S-map : {ty : U}
  {P Q : UUSet}(X : ∀{k v} → P k v → Q k v)
  → S P ty → S Q ty
S-map f (SX x)    = SX (f x)
S-map f Scp       = Scp
S-map f (Scns i xs) = Scns i (mapi f xs)
```

Computing a spine is easy, first we check whether or not x and y are equal. If they are, we are done. If not, we look at x and y as true sums of products and check if their constructors are equal, if they are, we zip the data together. If they are not, we zip x and y together and give up.

```

spine-cns : {ty : U}(x y : [ ty ]) → S Δs ty
spine-cns x y with sop x | sop y
spine-cns _ _ | strip cx dx | strip cy dy
    with cx  $\stackrel{?}{=}$  Fin cy
... | no _ = SX (inject cx dx , inject cy dy)
spine-cns _ _ | strip _ dx | strip cy dy
    | yes refl = Scns cy (zipp dx dy)

spine : {ty : U}(x y : [ ty ]) → S Δs ty
spine {ty} x y
    with dec-eq _  $\stackrel{?}{=}$  A _ ty x y
... | yes _ = Scp
... | no _ = spine-cns x y

zipp : {ty : Π}
    → [ ty ]p → [ ty ]p → Listl (λ k → Δs (α k) (α k)) ty
zipp {[]} _ _ = []
zipp {_ :: ty} (x , xs) (y , ys)
    = (i1 (x , unit) , i1 (y , unit)) :: zipp xs ys

```

The “application” relations specified by a spine $s = \text{spine } x \ y$, denoted s^b are defined by:

$$\begin{aligned}
\text{Scp}^b &= A \xleftarrow{id} A \\
(\text{SX } p)^b &= A \xleftarrow{p^b} A \\
(\text{Scns } i \ [s1 \ , \ \dots \ , \ sN])^b &= \Pi_k \Pi_j A_{kj} \xleftarrow{\text{inj}_i \cdot (s_1^b \times \dots \times s_n^b) \cdot \text{inj}_i^\circ} \Pi_k \Pi_j A_{kj}
\end{aligned}$$

where inj_i is the injection, with constructor i , into $\Pi_k T_k$. It corresponds to the relational lifting of function `injection`.

Note that, in the $(\text{SX } p)$ case, we simply ask for the “application” relation of p . The algorithm produces a $S \Delta_s$, so we have pairs on the leaves of the spine. In fact, either we have only one leaf or we have *arity* C_i leaves, where C_i is the common constructor of x and y in $\text{spine } x \ y$.

For a running example, let’s consider a datatype defined by:

```

2-3-TREE-F : σπ 1
2-3-TREE-F = []
    ⊕ (K kN) ⊗ I ⊗ I ⊗ []
    ⊕ (K kN) ⊗ I ⊗ I ⊗ I ⊗ []
    ⊕ []

```

We omit the `fz` for the `I` parts, as we only have one type variable. We also use $_ \oplus _$ and $_ \otimes _$ as aliases for $_ :: _$ with different precedences. As expected, there are three constructors:

```

2-node' 3-node' nil' : Constr 2-3-TREE-F
nil'      = fz
2-node'   = fs fz
3-node'   = fs (fs fz)

```

We can then consider a few spines over $[[2-3-TREE-F]]_{\text{Unit}}$ to illustrate the algorithm:

```

spine nil' (3-node' 10 unit unit unit) = SX (nil' , 3-node' 10 unit unit unit)
spine (2-node' 10 unit unit) (2-node' 15 unit unit) = Scns 2-node' [ (10 , 15) , (unit , unit) , (unit , unit) ]
spine nil' nil' = Scp

```

In the case where the spine is `Scp` or `Scns i` there is nothing left to be done and we have the best possible diff. Note that on the `Scns i` case we do *not* allow for rearranging of the parameters of the constructor *i*.

In the case where the spine is `SX`, we can do a better job! We can record which constructor changed into which and try to reconcile the data from both the best we can. Going one step at a time, let's first change one constructor into the other.

It is important to note that if the output of `spine` is a `SX`, then the constructors are *different*.

2.3 Constructor Changes

Let's take an example where the spine can not copy anything:

```
s = spine (2-node' 10 unit unit) (3-node' 10 unit unit unit)
      = SX (2-node' 10 unit unit , 3-node' 10 unit unit unit)
```

Here, we wish to say that we changed a `2-node'` into a `3-node'`. But we are then left with a problem about what to do with the data inside the `2-node'` and `3-node'`; this is where the notion of alignment will be in the picture. For now, we abstract it away by the means of a parameter, just like we did with the `S`. This time, however, we need something that receives products as inputs.

```
data C (P : IIISet) : U → U → Set where
  CX : {ty tv : U}
      → (i : Constr ty)(j : Constr tv)
      → P (typeOf ty i) (typeOf tv j)
      → C P ty tv
```

Note that `C` also makes up a functor, and hence can be mapped over:

```
C-map : {ty tv : U}
        {P Q : IIISet} (X : ∀ {k v} → P k v → Q k v)
        → C P ty tv → C Q ty tv
C-map f (CX i j x) = CX i j (f x)
```

Computing an inhabitant of `C` is trivial:

```
change : {ty tv : U} → [ ty ] → [ tv ] → C Δp ty tv
change x y with sop x | sop y
change _ _ | strip cx dx | strip cy dy = CX cx cy (dx , dy)
```

Now that we can compute change of constructors, we can refine our *s* above. We can compute `S-map change s` and we will have:

```
c = S-map change s
  = SX (CX 2-node' 3-node' ((10 , unit , unit) , (10 , unit , unit , unit)))
```

The “application” relation induced by `C` is trivial. We just need to pattern match, change the data of the constructor in whatever way we need, then inject into another type.

$$\begin{array}{ccc}
 V & \xleftarrow{(CX \ i \ j \ p)^b} & T \\
 \uparrow \text{inj}_j & & \downarrow \text{inj}_i^\circ \\
 \text{typeOf } V \ j & \xleftarrow{p^b} & \text{typeOf } T \ i
 \end{array}$$

Note that up until now, everything was deterministic! This is something we are bound to lose when talking about alignment.

2.4 Aligning Everything

On the literature for version control system, the *alignment* problem is the problem of mapping two strings l_1 and l_2 in \mathcal{L} into $\mathcal{L} \cup \{-\}$, for $\{-\} \not\subseteq \mathcal{L}$ such that the resulting strings l'_1 and l'_2 are of the same length such that for all i , it must not be happen that $l'_1[i] = - = l'_2[i]$. For example, Take strings $l_1 = \text{"CGTCG"}$ and $l_2 = \text{"GATAGT"}$, then, the following is an (optimal) alignment:

C	G	-	T	C	G	-
-	G	A	T	A	G	T

Let $\mathcal{DNA} = \{A, T, C, G\}$. Finding the table above is the same as finding a partial map:

$$f : \mathcal{DNA}^5 \rightarrow \mathcal{DNA}^6$$

such that $f(C, G, T, C, G) = (G, A, T, A, G, T)$. There are many ways of defining such a map. We would like, however, that our definition have a maximal domain, that is, we impose the least possible amount of restrictions. In this case, we can actually define f with some pattern matching as:

$$\begin{aligned} f(C, x, y, C, z) &= (x, A, y, A, z, T) \\ f \text{ } - &= \text{undefined} \end{aligned}$$

And it is easy to verify that, in fact, $f(C, G, T, C, G) = (G, A, T, A, G, T)$. Moreover, this is the *maximal* such f that still (provably) assigns the correct destination to the correct source.

On our running example, the leaf of c has type Δ_p (*typeOf 2-node'*) (*typeOf 3-node'*), and it's value is $((10, \text{unit}, \text{unit}), (10, \text{unit}, \text{unit}, \text{unit}))$. Note that we are now dealing with products of different arity. This step will let us say how to *align* one with the other!

On our example, as long as we align the 10 with the 10, the rest does not matter. One optimal alignment could be:

10	-	unit	unit
10	unit	unit	unit

2.4.1 Back to Agda

We will look at alignments from the “finding a map between products” perspective. Here is where our design space starts to grow, and so, we should start making some distinctions:

- We want to allow sharing. This means that there can be more than one variable in the defining pattern of our f .
- We do *not* allow permutations, as the search space would be too big. This means that the variables appear in the right-hand side of f in the same order as they appear in the left-hand-side.
- We do *not* allow contractions nor weakenings. That is, every variable on the left-hand-side of f must appear *exactly* once on the right-hand-side.

The following datatype describe such maps:

```
data Al (P : AASet) : Π → Π → Set where
  A0  : Al P [] []
  Ap1 : ∀{a ty tv} → [ a ]_a → Al P ty tv → Al P (a :: ty) tv
  Ap1° : ∀{a ty tv} → [ a ]_a → Al P ty tv → Al P ty (a :: tv)
  AX  : ∀{a a' ty tv} → P a a' → Al P ty tv → Al P (a :: ty) (a' :: tv)
```

Note that the indexes of `Al`, although represented as lists are, in fact, products. Well, turns out that lists and products are not so different after all. Let us represent the f we devised on the \mathcal{DNA} example using `Al`. Recall $f(C, x, y, C, z) = (x, A, y, A, z, T)$.

$$f \equiv \text{Ap1 } C \text{ (AX Scp (Ap1}^\circ A \text{ (AX Scp (AX (C , A) (AX Scp (Ap1}^\circ T \text{ A0))))))}$$

If we rename **Ap1** to *del*; **Ap1**[°] to *ins* and **AX** to *mod* we see some familiar structure arising! Aligning products is the same as computing the diff between heterogeneous lists! In fact, the **align** function is defined as:

```
align* : {ty tv : Π} → [ ty ]p → [ tv ]p → List (Al Δa ty tv)
align* {[]} {} m n = return A0
align* {[]} {v :: tv} m (n , nn)
  = Ap1° n <$> align* m nn
align* {y :: ty} {} (m , mm) n
  = Ap1 m <$> align* mm n
align* {y :: ty} {v :: tv} (m , mm) (n , nn)
  = AX (m , n) <$> align* mm nn
++ Ap1 m <$> filter (not ∘ is-ap1°) (align* mm (n , nn))
++ Ap1° n <$> filter (not ∘ is-ap1) (align* (m , mm) nn)
where
  is-ap1 : {ty tv : Π} → Al Δa ty tv → Bool
  is-ap1 (Ap1 _ _) = true
  is-ap1 _ = false

  is-ap1° : {ty tv : Π} → Al Δa ty tv → Bool
  is-ap1° (Ap1° _ _) = true
  is-ap1° _ = false
```

We are now doing things in the *List* monad. This is needed because there are many possible alignments between two products. For the moment, we refrain from choosing and compute all of them.

On another note, some of these alignments are simply dumb! We do not want to have both **Ap1** *x* (**Ap1**[°] *y* *a*) and **AX** (*x* , *y*) *a*. They are the same alignment. The **filters** are in charge of pruning out those branches from our search-space.

Sticking with our example, we can align the leaves of our *c* by computing the following expression, where **C-mapM** is simply the monadic variant of **C-map**.

```
a = C-mapM align* c
  = SX (CX 2-node' 3-node' (AX (10 , 10) (AX (unit , unit) ...)))
  :: SX (CX 2-node' 3-node' (Ap1 10 (AX (unit , 10) ...)))
  :: SX (CX 2-node' 3-node' (Ap1 10 (Ap1 unit ...)))
  :: SX (CX 2-node' 3-node' (Ap1° 10 (AX (10 , unit) ...)))
  ...
```

Now we have a problem. Which of the patches above should we chose to be *the* patch? Recall that we mentioned we wanted to find the alignment with *maximum domain*. Something interesting happens if we look at patches from their “application” relation, but first, we define the “application” relations of **Al**:

$$\begin{aligned} (\text{AX } a_1 a_2)^b &= B \times \Pi D \xleftarrow{a_1^b \times a_2^b} A \times \Pi C \\ (\text{Ap1 } x a)^b &= \Pi B \xleftarrow{\langle \underline{x}, a^b \rangle^\circ} X \times \Pi A \\ (\text{Ap1}^\circ x a)^b &= X \times \Pi B \xleftarrow{\langle \underline{x}, a^b \rangle} \Pi A \\ \text{A0}^b &= \mathbb{1} \xleftarrow{\top} \mathbb{1} \end{aligned}$$

3 Patches as Relations

In order to better illustrate this concept, we need a simpler example first. Let's consider the following type with no type variables:

```
Type1 = K kN ⊗ []
        ⊕ K kN ⊗ K kN ⊗ []
        ⊕ []
```

It clearly has two constructors:

```
C1 C2 : Constr Type1
C1     = fz
C2     = fs fz
```

Now, let's take two inhabitants of `Type1`.

```
x y : [[ Type1 ]]
x    = inject C2 (4 , 10 , unit)
y    = inject C1 (10 , unit)
```

There are two possible options for `diff x y`:

```
ds : Patch* Type1
ds = SX (CX C2 C1 (AX (4 , 10) (Ap1 10 A0))) - P1
    :: SX (CX C2 C1 (Ap1 4 (AX (10 , 10) A0))) - P2
    :: []
```

Consider the semantics for Δ as described in the discussion box at Section 2.1, that is,

$$(x, y)^b = \begin{cases} id & \text{if } x \equiv y \\ \underline{y} \cdot \underline{x}^\circ & \text{otherwise} \end{cases}$$

Then it becomes clear that we want to select patch (P2) instead of (P1). In fact, there is a deeper underlying reason for that! Looking at the two patches as relations (after some simplifications), we have:

$$\begin{aligned} P1^b &= \text{inj}_{C_1} \cdot \langle (\underline{4} \cdot \underline{10}^\circ), \underline{10} \rangle^\circ \cdot \text{inj}_{C_2}^\circ \\ P2^b &= \text{inj}_{C_1} \cdot \langle \underline{4}, id \rangle^\circ \cdot \text{inj}_{C_2}^\circ \end{aligned}$$

Drawing them in a diagram we have:

$$\begin{array}{ccc} \text{typeOf}_{\text{Type1}} C_2 \equiv \mathbb{N} \times \mathbb{N} & \xleftarrow{\text{inj}_{C_2}^\circ} & [[\text{Type1}]] \\ \langle \underline{4}, id \rangle^\circ \downarrow \downarrow \langle \underline{4} \cdot \underline{10}^\circ, \underline{10} \rangle^\circ & & \begin{array}{c} \downarrow \downarrow \\ P2^b \mid \mid P1^b \\ \downarrow \downarrow \end{array} \\ \text{typeOf}_{\text{Type1}} C_1 \equiv \mathbb{N} & \xrightarrow{\text{inj}_{C_1}} & [[\text{Type1}]] \end{array}$$

Here, we have something curious going on... We have that $P1^b \subseteq P2^b$. To see this is not very hard. First, composition and converses are monotonous with respect to \subseteq . We are left to check that:

$$\begin{aligned} &\langle \underline{4} \cdot \underline{10}^\circ, \underline{10} \rangle \subseteq \langle \underline{4}, id \rangle \\ &\equiv \{ \text{split universal} \} \\ &\pi_1 \cdot \langle \underline{4} \cdot \underline{10}^\circ, \underline{10} \rangle \subseteq \underline{4} \wedge \pi_2 \cdot \langle \underline{4} \cdot \underline{10}^\circ, \underline{10} \rangle \subseteq id \end{aligned}$$

The first proof obligation is easy to calculate with:

$$\begin{aligned}
& \pi_1 \cdot < \underline{4} \cdot \underline{10}^\circ, \underline{10} > \subseteq \underline{4} \\
& \Leftarrow \{ \pi_1\text{-cancel} ; \subseteq\text{-trans} \} \\
& \underline{4} \cdot \underline{10}^\circ \subseteq \underline{4} \\
& \Leftarrow \{ \text{Leibniz} \} \\
& \underline{4} \cdot \underline{10}^\circ \cdot \underline{10} \subseteq \underline{4} \cdot \underline{10} \\
& \equiv \{ \underline{a}^\circ \cdot \underline{a} \equiv \top \} \\
& \underline{4} \cdot \top \subseteq \underline{4} \cdot \underline{10} \\
& \equiv \{ \underline{a} \cdot \underline{b} \equiv \underline{a} \} \\
& \underline{4} \cdot \top \subseteq \underline{4} \\
& \equiv \{ \underline{a} \cdot \top \equiv \underline{a} \} \\
& \underline{4} \subseteq \underline{4} \\
& \equiv \{ \subseteq\text{-refl} \} \\
& \text{True}
\end{aligned}$$

The second is easier to prove once we add variables!

$$\begin{aligned}
& \pi_2 \cdot < \underline{4} \cdot \underline{10}^\circ, \underline{10} > \subseteq \text{id} \\
& \equiv \{ \text{Add variables} \} \\
& \forall x, y . x (\pi_2 \cdot < \underline{4} \cdot \underline{10}^\circ, \underline{10} >) y \Rightarrow x = y \\
& \equiv \{ \text{PF expand composition} \} \\
& \forall x, y . \exists z . x (\pi_2) z \wedge z < \underline{4} \cdot \underline{10}^\circ, \underline{10} > y \Rightarrow x = y \\
& \equiv \{ \text{Types force } z = (z_1, z_2) \} \\
& \forall x, y . \exists z_1, z_2 . x (\pi_2) (z_1, z_2) \wedge (z_1, z_2) < \underline{4} \cdot \underline{10}^\circ, \underline{10} > y \Rightarrow x = y \\
& \equiv \{ \pi_2 \text{ def} \} \\
& \forall x, y . \exists z_1, z_2 . x = z_2 \wedge (z_1, z_2) < \underline{4} \cdot \underline{10}^\circ, \underline{10} > y \Rightarrow x = y \\
& \equiv \{ \text{split def} \} \\
& \forall x, y . \exists z_1, z_2 . x = z_2 \wedge z_1 (\underline{4} \cdot \underline{10}^\circ) y \wedge z_2 (\underline{10}) y \Rightarrow x = y \\
& \equiv \{ \text{points def} \} \\
& \forall x, y . \exists z_1, z_2 . x = z_2 \wedge z_1 = 4 \wedge y = 10 \wedge z_2 = 10 \Rightarrow x = y \\
& \equiv \{ \text{substitutions ; weakening} \} \\
& \forall x, y . \exists z_2 . x = 10 \wedge y = 10 \Rightarrow x = y \\
& \equiv \{ \text{trivial} \} \\
& \text{True}
\end{aligned}$$

Nevertheless, it is clear which patch we should choose! We should always choose the patch that gives rise to the biggest relation, as this is applicable to much more elements.

Hence, our *cost* functions will count how many elements of the domain and range of the “application” relation of a patch are *fixed*. Note that the **S** and **C** parts of the algorithm are completely deterministic, hence they should *not* contribute to cost:

$$\begin{aligned}
& \text{S-cost} : \{ty : \mathbf{U}\} \{P : \mathbf{UUSet}\} (doP : \{k v : \mathbf{U}\} \rightarrow P \ k \ v \rightarrow \mathbf{N}) \\
& \quad \rightarrow \mathbf{S} \ P \ ty \rightarrow \mathbf{N} \\
& \text{S-cost } doP \ (\mathbf{SX} \ x) = doP \ x \\
& \text{S-cost } doP \ \mathbf{Scp} = 0 \\
& \text{S-cost } doP \ (\mathbf{Scns} \ i \ xs) = \text{foldr}_i (\lambda \ h \ r \rightarrow doP \ h + r) \ 0 \ xs \\
& \text{C-cost} : \{ty \ tv : \mathbf{U}\} \{P : \mathbf{IIISet}\} (doP : \{k v : \mathbf{II}\} \rightarrow P \ k \ v \rightarrow \mathbf{N}) \\
& \quad \rightarrow \mathbf{C} \ P \ ty \ tv \rightarrow \mathbf{N} \\
& \text{C-cost } doP \ (\mathbf{CX} \ i \ j \ x) = doP \ x
\end{aligned}$$

An **Alignment** might fix one element on the source, using **Ap1** or one element on the destination, using **Ap1°**.

```

Al-cost : {ty tv : Π} {P : AASet} (doP : {k v : Atom} → P k v → ℕ)
  → Al P ty tv → ℕ
Al-cost doP A0 = 0
Al-cost doP (Ap1 x a) = 1 + Al-cost doP a
Al-cost doP (Ap1° x a) = 1 + Al-cost doP a
Al-cost doP (AX x a) = doP x + Al-cost doP a

```

Last but not least, a **Δ** will either fix 2 elements: one in the source that becomes one in the destination; or none, when we just copy the source.

```

cost-delta-raw : ℕ
cost-delta-raw = 2

cost-delta : ∀ {α} {A : Set α} {ty tv : A} (P : A → Set)
  (eqA : (x y : A) → Dec (x ≡ y))
  (eqP : (k : A) (x y : P k) → Dec (x ≡ y))
  → delta P ty tv → ℕ
cost-delta {ty = ty} {tv = tv} P eqA eqP (pa1 , pa2)
  with eqA ty tv
... | no _ = cost-delta-raw
cost-delta {ty = ty} P eqA eqP (pa1 , pa2)
  | yes refl with eqP ty pa1 pa2
... | no _ = cost-delta-raw
... | yes _ = 0

```

According to these definitions, the cost of (P1) above is 3, where the cost of (P2) is 1.

3.1 Patches for Regular Types

Now that we have *spines*, *changes* and *alignments* figured out, we can define a patch as:

```

Patch : AASet → U → Set
Patch P = S (C (Al P))

```

Computing inhabitants of such type is done with:

```

diff1* : {ty : U} (x y : [[ ty ]]) → Patch* ty
diff1* x y = S-mapM (C-mapM (uncurry align*) ∘ uncurry change) (spine x y)

```

where **Patch*** is defined as **List** (**Patch** **Δ_a**).

3.2 Conjectures About the **cost** function

Here we conjecture a few lemmas about the interplay of the cost function and the “application” relation. Let *P*, *Q* and *R* be patches.

- i) If *P* has a lower cost than *Q*, then the domain and range of the “application” relation of *P* contains the “application” relation of *Q*.

$$\text{cost } P < \text{cost } Q \Rightarrow Q^b \subseteq P^b \cdot \top \cdot P^b$$

Needs discussion:

This is not as simple as

$$\text{cost } P < \text{cost } Q \Rightarrow Q^b \subseteq P^b$$

Take two *Deltas*, $px = (10, 50)$ and $py = (30, 30)$. Trivially, $\text{cost } py = 0$ and $\text{cost } px = 2$. Now, $px^b = \underline{50} \cdot \underline{10}^\circ$ and $py^b = id$. It is not true that $\underline{50} \cdot \underline{10}^\circ \subseteq id$!

If we state, however: Let P, Q in $\text{diff} * x y$; $\text{cost } P < \text{cost } Q \Rightarrow Q^b \subseteq P^b$. Seems more likely. As the above counter example would not work anymore. $\text{diff} * 10 \ 50 = (10, 50) :: []$.

- ii) If P and Q have equal cost, it means that there is at least one place where P and Q are doing *the same thing*, hence there is a patch that copies this *same thing* and costs strictly less.

$$\text{cost } P \equiv \text{cost } Q \Rightarrow \exists R \cdot \text{cost } R < \text{cost } P$$